Stability of equilibrium points

Higher-Order Scalar Difference Equations

Def. A constant solution (x_n), $x_n \equiv x^*$ of the difference equation

$x_{n+k} = f(x_{n+k-1}, \ldots, x_n)$

is called an *equilibrium solution.*
A point x* is called an *equilibrium point* for the difference equation and it is a solution of the equation
x* = f(x*,...,x*)

Linear case
Consider the k-order difference equation

$x_{n+k} + p_1 x_{n+k-1} + p_2 x_{n+k-2} + \cdots + p_k x_n = 0$

x* = 0

the characteristic equation is:

$q^k + p_1 q^{k-1} + p_2 q^{k-2} + \cdots + p_k = 0$

Theorem.
i) x* = 0 is asymptotically stable if and only if $|q| < 1$ for all roots of characteristic equation.
ii) x* = 0 is locally stable if and only if $|q| \le 1$ for all roots of characteristic equation, $|q| = 1$ holds for simple roots.
iii) x* = 0 is unstable if and only if there exists a root q of the characteristic equation for which $|q| > 1$ or $|q| = 1$ and q is not a simple root.

```
> restart;
> eqd:=6*x(n+2)+x(n+1)-2*x(n)=0;
```

$$eqd := 6\,x(n+2) + x(n+1) - 2\,x(n) = 0$$

```
> ech:=6*q^2+q-2=0;
```

$$ech := 6\,q^2 + q - 2 = 0$$

```
> solve(ech,q);
```

$$\frac{1}{2}, -\frac{2}{3}$$

```
> rsolve(eqd,x(n));
```

$$\frac{1}{3}\left(\frac{9}{7}\,x(0) - \frac{18}{7}\,x(1)\right)\left(-\frac{2}{3}\right)^n - \frac{1}{2}\left(-\frac{8}{7}\,x(0) - \frac{12}{7}\,x(1)\right)\left(\frac{1}{2}\right)^n$$

```
> x[0]:=20;x[1]:=40;N:=50;
```

$$x_0 := 20$$
$$x_1 := 40$$
$$N := 50$$

```
> for i from 0 to N-2 do
```

```
    x[i+2]:=1/6*(-x[i+1]+2*x[i])
  end do:
> plot([[n,x[n]]$n=0..N],style=point,symbol=circle);
```



```
> restart;
> eqd:=x(n+2)-6*x(n+1)+18*x(n)=0;
```

$$eqd := x(n+2) - 6\,x(n+1) + 18\,x(n) = 0$$

```
> ech:=q^2-6*q+18=0;
```

$$ech := q^2 - 6\,q + 18 = 0$$

```
> qq:=solve(ech,q);
```

$$qq := 3 + 3\,\mathrm{I},\, 3 - 3\,\mathrm{I}$$

```
> abs(qq[1]);
```

$$3\sqrt{2}$$

```
> sol:=rsolve(eqd,x(n));
```

$$sol := \left(\frac{1}{12} - \frac{1}{12}\,\mathrm{I}\right)\left(6\,\mathrm{I}\,x(0) - \mathrm{I}\,x(1) + x(1)\right)(3+3\,\mathrm{I})^n + \left(-\frac{1}{12} - \frac{1}{12}\,\mathrm{I}\right)\left(6\,\mathrm{I}\,x(0) - \mathrm{I}\,x(1) - x(1)\right)(3-3\,\mathrm{I})^n$$

```
> evalc(sol);
```

$$x(0)\, e^{\frac{1}{2} n \ln(18)} \cos\left(\frac{1}{4} n \pi\right) - \left(-\frac{1}{6} x(1) + \frac{1}{2} x(0)\right) e^{\frac{1}{2} n \ln(18)} \sin\left(\frac{1}{4} n \pi\right) + \left(\frac{1}{6} x(1) - \frac{1}{2} x(0)\right) e^{\frac{1}{2} n \ln(18)} \sin\left(\frac{1}{4} n \pi\right) + \mathrm{I}\left(\left(-\frac{1}{6} x(1) + \frac{1}{2} x(0)\right) e^{\frac{1}{2} n \ln(18)} \cos\left(\frac{1}{4} n \pi\right) + \left(\frac{1}{6} x(1) - \frac{1}{2} x(0)\right) e^{\frac{1}{2} n \ln(18)} \cos\left(\frac{1}{4} n \pi\right)\right)$$

```
> simplify(evalc(sol));
```

$$\frac{1}{3} e^{\frac{1}{2} n (\ln(2) + 2 \ln(3))} \left(3\, x(0) \cos\left(\frac{1}{4} n \pi\right) - 3 \sin\left(\frac{1}{4} n \pi\right) x(0) + \sin\left(\frac{1}{4} n \pi\right) x(1)\right)$$

```
> simplify(evalc(sol), exp);
```

$$x(0)\, 18^{\frac{1}{2} n} \cos\left(\frac{1}{4} n \pi\right) - \left(-\frac{1}{6} x(1) + \frac{1}{2} x(0)\right) 18^{\frac{1}{2} n} \sin\left(\frac{1}{4} n \pi\right) + \left(\frac{1}{6} x(1) - \frac{1}{2} x(0)\right) 18^{\frac{1}{2} n} \sin\left(\frac{1}{4} n \pi\right) + \mathrm{I}\left(\left(-\frac{1}{6} x(1) + \frac{1}{2} x(0)\right) 18^{\frac{1}{2} n} \cos\left(\frac{1}{4} n \pi\right) + \left(\frac{1}{6} x(1) - \frac{1}{2} x(0)\right) 18^{\frac{1}{2} n} \cos\left(\frac{1}{4} n \pi\right)\right)$$

```
> x[0]:=0.1;x[1]:=-0.3;N:=17;
```

$$x_0 := 0.1$$
$$x_1 := -0.3$$
$$N := 17$$

```
> for i from 0 to N-2 do
    x[i+2]:=6*x[i+1]-18*x[i]
  end do:
> plot([[n,x[n]]$n=0..N],style=point,symbol=circle);
```



Nonlinear case

$x_{n+k} = \mathrm{f}(x_{n+k-1},\dots,x_n)$

$f = f(u_1, \dots, u_k)$

$p_i = \dfrac{\partial}{\partial u_i} f(\mathrm{x}^*,\dots,\mathrm{x}^*)$

linearized equation in (x*, x*, . . . , x*)

$y_{n+k} = p_1 y_{n+k-1} + p_2 y_{n+k-2} + \cdots + p_k y_n$

characteristic equation corresponding to the linearized equation is

$q^k = p_1 q^{k-1} + p_2 q^{k-2} + \cdots + p_k$

Theorem (The Linearized Stability Theorem).

Suppose that f is continuously differentiable on an open neighborhood G from R^{k+1} of (x*, x*, . . . , x*), where x* is an equilibrium point of the nonlinear difference equation. Then the following statements are true:
(i) If all the characteristic roots of the linearized characteristic equation lie inside the unit disk in the complex plane ($|q| < 1$), then the equilibrium point x* is locally asymptotically stable.
(ii) If at least one characteristic root of the linearized characteristic equation is outside the unit disk ($|q| > 1$) in the complex plane, the equilibrium point x* is unstable.
(iii) If one characteristic root is on the unit disk and all the other characteristic roots are either inside or on the unit disk, then the equilibrium point x* may be stable, unstable, or asymptotically stable.

The Schur-Cohn Criterion
(a) for the equation the equation

$$q^2 = p_1 q + p_2$$

we have:
$|q_{1,2}| < 1$ if and only if $|p_1| < 1 - p_2 < 2$

(b) for the equation the equation

$$q^3 = p_1 q^2 + p_2 q + p_3$$

we have:
$|q_{1,2,3}| < 1$ if and only if $|p_3 + p_1| < 1 - p_2$ and $|p_1 p_3 + p_2| < 1 - p_3^2$

Example

The second order Beverton-Holt model

$$x_{n+1} = \frac{r K \left(\alpha x_n + \beta x_{n-1}\right)}{K + (r-1) x_{n-1}}$$

has been used to model populations of bottom-feeding fish, where $\alpha + \beta = 1$, r > 0, K > 0. These species have very high fertility rates and very low survivorship to adulthood. Furthermore, recruitment is essentially unaffected by fishing. In this model, the future generation x_{n+1} depends not only on the present generation x_n but also on the previous generation x_{n-1}.

```
> restart:
> f:=(u,v)->r*K*(alpha*u+beta*v)/(K+(r-1)*v);
```

$$f := (u, v) \rightarrow \frac{r K (\alpha u + \beta v)}{K + (r-1) v}$$

```
> eq:=x=f(x,x);
```

$$eq := x = \frac{r K (\alpha x + \beta x)}{K + (r-1) x}$$

```
> eqp:=solve(eq,x);
```

$$eqp := 0, \frac{K (\alpha r + \beta r - 1)}{r - 1}$$

```
> x1:=eqp[1];x2:=simplify(subs(beta=1-alpha,eqp[2]));
```

$$x1 := 0$$
$$x2 := K$$

```
> p1:=D[1](f)(0,0);
```

$$p1 := \alpha r$$

```
> p2:=D[2](f)(x1,x1);
```

$$p2 := \beta r$$

```
> lineq:=y(n+1)=p1*y(n)+p2*y(n-1);
```

$$lineq := y(n+1) = \alpha r y(n) + \beta r y(n-1)$$

```
> chareq:=q^2=p1*q+p2;
```

$$chareq := q^2 = \alpha q r + \beta r$$

```
> rr:=solve(chareq,q);
```

$$rr := \frac{1}{2}\alpha r + \frac{1}{2}\sqrt{\alpha^2 r^2 + 4\beta r},\ \frac{1}{2}\alpha r - \frac{1}{2}\sqrt{\alpha^2 r^2 + 4\beta r}$$

```
> abs(subs(beta=1-alpha,rr[1]));
```

$$\left|\frac{1}{2}\alpha r + \frac{1}{2}\sqrt{\alpha^2 r^2 + 4(1-\alpha) r}\right|$$

```
> abs(p1)<1-p2;
```

$$|\alpha r| < -\beta r + 1$$

```
> 1-p2<2
```

$$-\beta r < 1$$

Analysing the roots x_1*=0 is is locally asymptotically stable if $0 < r < 1$.

```
> p1:=D[1](f)(x2,x2);
```

$$p1 := \frac{r K \alpha}{K + (r-1) K}$$

```
> p1:=simplify(p1);
```

$$p1 := \alpha$$

```
> p2:=D[2](f)(x2,x2);
```

$$p2 := \frac{r K \beta}{K + (r-1) K} - \frac{r K (K\alpha + K\beta)(r-1)}{(K + (r-1) K)^2}$$

```
> p2:=simplify(p2);
```

$$p2 := -\frac{\alpha r - \alpha - \beta}{r}$$

```
> p2:=subs(beta=1-alpha,p2);
```

$$p2 := -\frac{\alpha r - 1}{r}$$

```
> lineq:=y(n+1)=p1*y(n)+p2*y(n-1);
```

$$lineq := y(n+1) = \alpha y(n) - \frac{(\alpha r - 1) y(n-1)}{r}$$

```
> chareq:=q^2=p1*q+p2;
```

$$chareq := q^2 = \alpha\, q - \frac{\alpha\, r - 1}{r}$$

```
> solve(chareq,q);
```

$$\frac{1}{2}\,\frac{\alpha\, r + \sqrt{\alpha^2\, r^2 - 4\,\alpha\, r^2 + 4\, r}}{r},\ -\frac{1}{2}\,\frac{-\alpha\, r + \sqrt{\alpha^2\, r^2 - 4\,\alpha\, r^2 + 4\, r}}{r}$$

```
> abs(p1)<1-p2;
```

$$|\alpha| < 1 + \frac{\alpha\, r - 1}{r}$$

```
> 1-p2<2
```

$$\frac{\alpha\, r - 1}{r} < 1$$

Analysing the roots we get that the equilibrium $x_2^* = K$ is locally asymptotically stable if r > 1.

Remark. If r > 1 then $x_2^* = K$ is globally asymptotically stable

Numerical simulations

```
> K:=100;r:=0.8;alpha:=0.5;beta:=1-alpha;N:=100;
```

$$K := 100$$
$$r := 0.8$$
$$\alpha := 0.5$$
$$\beta := 0.5$$
$$N := 100$$

```
> f(u,v);
```

$$\frac{80.0\,(0.5\, u + 0.5\, v)}{100 - 0.2\, v}$$

```
> x[0]:=20;x[1]:=40;
```

$$x_0 := 20$$
$$x_1 := 40$$

```
> for i from 0 to N-2 do
    x[i+2]:=f(x[i+1],x[i])
  end do:
> plot([[n,x[n]]$n=0..N],style=point,symbol=circle);
```




```
> K:=100;r:=1.2;alpha:=0.5;beta:=1-alpha;N:=100;
```

$$K := 100$$
$$r := 1.2$$
$$\alpha := 0.5$$
$$\beta := 0.5$$

$$N := 100$$

```
> f(u,v);
```

$$\frac{120.0\,(0.5\,u + 0.5\,v)}{100 + 0.2\,v}$$

```
> x[0]:=10;x[1]:=20;
```

$$x_0 := 10$$
$$x_1 := 20$$

```
> for i from 0 to N-2 do
    x[i+2]:=f(x[i+1],x[i])
  end do:
```

```
> plot([[[n,x[n]]$n=0..N],[[n,K]$n=0..N]],style=[point,point],
  symbol=[circle,cross],color=[red,green]);
```




```
> restart;
```

Autonomous Systems of Difference equations

Linear case

$x_{n+1} = A\, x_n$

the solution

$x_n = A^n\, x_0$

Theorem. The following statements hold:
(i) The zero solution is locally stable if and only if $|\lambda| \le 1$ and the eigenvalues of unit modulus are simple.
(ii) The zero solution is asymptotically stable if and only if $|\lambda| < 1$ for all eigenvalues of A.

Example

$$\begin{cases} x_{n+1} = \frac{1}{2} x_n + y_n \\ y_{n+1} = \frac{1}{3} x_n + \frac{1}{8}\, y_n \end{cases}$$

```
> restart:
> with(linalg):
> A:=matrix([[1/2,1],[1/3,1/8]]);
```

$$A := \begin{bmatrix} \frac{1}{2} & 1 \\ \frac{1}{3} & \frac{1}{8} \end{bmatrix}$$

```
> eigenvals(A);
```

$$\frac{5}{16} + \frac{1}{48}\sqrt{849},\ \frac{5}{16} - \frac{1}{48}\sqrt{849}$$

```
> evalf(%);
```

$$0.9195334284,\ -0.2945334284$$

both eigenvalues $|\lambda_{1,2}| < 1$, so (0,0) is locally asymptotically stable

```
> N:=100;x[0]:=50;y[0]:=-20;
```

$$N := 100$$
$$x_0 := 50$$
$$y_0 := -20$$

```
> for i from 0 to N-1 do
    x[i+1]:=1/2*x[i]+y[i];
    y[i+1]:=1/3*x[i]+1/8*y[i];
  end do:
> plot([[[n,x[n]]$n=0..N],[[n,y[n]]$n=0..N]],style=[point,
```

```
point],symbol=[circle, cross], color=[red,blue]);
```



Nonlinear case

$x_{n+1} = f(x_n)$

x*=f(x*) equilibrium point

linearized system

y_{n+1} = Jacobian(f) (x*) y_n

Theorem. The following statements hold:
(i) The x* is locally asymptotically stable if and only if $|\lambda| < 1$ for all eigenvalues of the Jacobian(f) (x*).
(ii) The x* is unstable if there exists an eigenvalue of the Jacobian(f) (x*) such that $|\lambda| > 1$.

Example

$$\begin{cases} x_{n+1} = y_n \\ y_{n+1} = \dfrac{\alpha\, y_n}{1+\beta\, x_n} \end{cases}$$

```
> restart;with(linalg):
> f1:=(x,y)->y;
```

$$f1 := (x, y) \to y$$

```
> f2:=(x,y)->alpha*y/(1+beta*x);
```

$$f2 := (x, y) \to \frac{\alpha\, y}{1+\beta\, x}$$

```
> eqp:=solve({f1(x,y)=x,f2(x,y)=y},{x,y});
```

$$eqp := \{x=0, y=0\}, \left\{x = \frac{\alpha-1}{\beta}, y = \frac{\alpha-1}{\beta}\right\}$$

```
> J:=jacobian([f1(x,y),f2(x,y)],[x,y]);
```

$$J := \begin{bmatrix} 0 & 1 \\ -\dfrac{\alpha\, y\, \beta}{(\beta\, x+1)^2} & \dfrac{\alpha}{\beta\, x+1} \end{bmatrix}$$

```
> A1:=subs(x=0,y=0,evalm(J));
```

$$A1 := \begin{bmatrix} 0 & 1 \\ 0 & \alpha \end{bmatrix}$$

```
> eigenvals(A1);
```

$$0, \alpha$$

```
> A2:=subs(x=(alpha-1)/beta,y=(alpha-1)/beta,evalm(J));
```

$$A2 := \begin{bmatrix} 0 & 1 \\ -\dfrac{\alpha-1}{\alpha} & 1 \end{bmatrix}$$

```
> eigenvals(A2);
```

$$\frac{1}{2}\,\frac{\alpha+\sqrt{-3\,\alpha^2+4\,\alpha}}{\alpha}, -\frac{1}{2}\,\frac{-\alpha+\sqrt{-3\,\alpha^2+4\,\alpha}}{\alpha}$$

```
> charpoly(A2,lambda);
```

$$\frac{\alpha\,\lambda^2-\alpha\,\lambda+\alpha-1}{\alpha}$$

Numerical simulation

```
> x[0]:=10;y[0]:=20;
```

$$x_0 := 10$$
$$y_0 := 20$$

```
> alpha:=0.8;beta:=1;N:=100;
```

$$\alpha := 0.8$$
$$\beta := 1$$
$$N := 100$$

```
> for i from 0 to N-1 do
    x[i+1]:=f1(x[i],y[i]);
    y[i+1]:=f2(x[i],y[i]);
  end do:
> plot([[n,x[n]]$n=0..N],style=point,symbol=circle);
```




```
> plot([[n,y[n]]$n=0..N],style=point,symbol=circle);
```




```
> x[0]:=10;y[0]:=15;
```

$$x_0 := 10$$
$$y_0 := 15$$

```
> alpha:=3;beta:=0.3;N:=100;
```

$$\alpha := 3$$
$$\beta := 0.3$$
$$N := 100$$

```
> xx:=(alpha-1)/beta;
```

$$xx := 6.666666667$$

```
> for i from 0 to N-1 do
    x[i+1]:=f1(x[i],y[i]);
    y[i+1]:=f2(x[i],y[i]);
  end do:
> plot([[[n,x[n]]$n=0..N],[[n,xx]$n=0..N]],style=[point,point],
  symbol=[circle,cross],color=[red,blue]);
```




```
> plot([[[n,y[n]]$n=0..N],[[n,xx]$n=0..N]],style=[point,point],
  symbol=[circle,cross],color=[red,blue]);
```



14
12
10
8
6
0
20
40
60
80
100