Solving Difference equations with MAPLE

MAPLE can solve some difference equations using **`rsolve`** command. The command structure is
rsolve(**eqns**, **fcns**)
where

- The function **rsolve** attempts to solve the recurrence relation(s) specified in **eqns** for the functions in **fcns**, returning an expression for the general term of the function.
- The first argument should be a single recurrence relation or a set of recurrence relations and boundary conditions. Any expressions in **eqns** which are not equations will be understood to be equal to zero.
- The second argument **fcns** indicates the functions that **rsolve** should solve for. A function may be represented by a name or an unevaluated function call whose argument is a name. If the function is represented by a name, the index variable for the solution is deduced from the recurrence equations. If the function is represented by an unevaluated function call, the argument name is used as the solution index. If the solution index differs from the index name in a recurrence equation, then the equation must be independent of the solution index.
- If **fcns** is a set, the solution is returned as a set of equations. The left hand side of each equation will be the function name. Otherwise, an expression for the solution is returned.
- The procedure **rsolve** can currently solve linear recurrences with constant coefficients, systems of linear recurrences with constant coefficients, divide and conquer recurrences with constant coefficients, many first order linear recurrences, and some nonlinear first order recurrences.
- If insufficient boundary conditions are given, **rsolve** uses symbolic function names as default values.

For linear recurrences with constant coefficients and systems of linear recurrences with constant coefficients, if no boundary conditions are given, symbolic function calls starting with index 0 are used.

- Boundary conditions for linear recurrences with constant coefficients and systems of linear recurrences with constant coefficients may be specified in one of three ways:

```
t(a)=c,         t(a..b)=c,         t(k=a..b)=f(k),
```

where **a** and **b** are integers, **c** is a constant with respect to the index variable for the sequence, **k** is a name, and **f(k)** is an expression of **k**. The boundary condition **t(a..b)=c** is equivalent to **t(a)=t(a+1)=...=t(b)=c** and the condition **t(k=a..b)=f(k)** is equivalent to **t(a)=f(a), t(a+1)=f(a+1), ..., t(b)=f(b)**,

First order linear difference equations

$y_{n+1} + a\, y_n = b_n$ nonhomogeneous first order difference equation

$y_{n+1} + a\, y_n = 0$ homogeneous first order difference equation

General solution
$y_n = c\, y\_hom_n + y\_p_n$

where
y_hom_n is the general solution of the homogeneous equation

y_p_n is a particular solution of the nonhomogeneous equation

The case of homogeneous equation

$y_{n+1} + a\, y_n = 0$

If we want to find solution of the form $y_n = q^n$ then q must be a solution of the equation $q + a = 0$ which is called the characteristic equation. The general solution of the homogeneous equation is generated by the root of the characteristic equation $q = -a$.
General solution: $y\_hom_n = c\,(-a)^n$
The constant c can be found from initial condition $y_0 = \alpha$

```
> eq1:=y(n+1)+a*y(n)=0;
```

$$eq1 := \mathrm{y}(n+1) + a\,\mathrm{y}(n) = 0$$

```
> rsolve(eq1,y(n));
```

$$\mathrm{y}(0)\,(-a)^n$$

```
> rsolve({eq1,y(0)=alpha},y(n));
```

$$\alpha\,(-a)^n$$

The case of nonhomogeneous equation
General solution
$y_n = y\_hom_n + y\_p_n$

$y\_hom_n = y_0\,(-a)^n$ is the general solution of the homogeneous equation
y_p_n , the particular solution of the nonhomogeneous equation, generaly, can be found using the variation of the constant method, we try to find a particular solution of the form
$y\_p_n = c_n\,(-a)^n$, we try to find the sequence (c_n) from the condition that (y_p_n) to be a solution of the nonhomogeneous equation

So, $y\_p_{n+1} + a\, y\_p_n = b_n$
$c_{n+1}\,(-a)^{(n+1)} - (-a)\,c_n\,(-a)^n = b_n$
$c_{n+1} - c_n = (-a)^{(-(n+1))}\,b_n$

$n = 0 \qquad c_1 - c_0 = (-a)^{(-1)}\,b_0$
$n = 1 \qquad c_2 - c_1 = (-a)^{(-2)}\,b_1$
....
$n = n - 2 \qquad c_{n-1} - c_{n-2} = (-a)^{(-(n-1))}\,b_{n-2}$

$n = n - 1 \qquad c_n - c_{n-1} = (-a)^{(-n)}\, b_{n-1}$

$$c_n - c_0 = \sum_{k=1}^{n-1} (-a)^{(-k)}\, b_{k-1}$$

choosing $c_0 = 0$, we can write

$$c_n = \sum_{k=1}^{n-1} (-a)^{(-k)}\, b_{k-1}$$

Thus, the particular solution has the following form

$$y\_p_n = (-a)^n\, c_n \;=\; (-a)^n \left(\sum_{k=1}^{n-1} (-a)^{(-k)}\, b_{k-1} \right)$$

$$y\_p_n = \sum_{k=1}^{n-1} (-a)^{(n-k)}\, b_{k-1}$$

```
> eq1_nonhom:=y(n+1)+a*y(n)=b(n);
```

$$eq1\_nonhom := \mathrm{y}(n+1) + a\,\mathrm{y}(n) = \mathrm{b}(n)$$

```
> rsolve(eq1_nonhom,y(n));
```

$$\mathrm{y}(0)\,(-a)^n + \left(\sum_{n0=1}^{n} (-a)^{(n-n0)}\, \mathrm{b}(n0-1) \right)$$

Particular cases for (b_n)

1. $b_n = b$ for all n

```
> eq1_nonhom:=y(n+1)+a*y(n)=b;
```

$$eq1\_nonhom := \mathrm{y}(n+1) + a\,\mathrm{y}(n) = b$$

```
> rsolve(eq1_nonhom,y(n));
```

$$\mathrm{y}(0)\,(-a)^n + \frac{b}{1+a} - \frac{b\,(-a)^n}{1+a}$$

2. $b_n = P_m(n)$ - polinomial of degree m with respect to n

```
> eq1_nonhom:=y(n+1)+a*y(n)=n^2+n;
```

$$eq1\_nonhom := \mathrm{y}(n+1) + a\,\mathrm{y}(n) = n^2 + n$$

```
> rsolve(eq1_nonhom,y(n));
```

$$y(0)(-a)^n + \frac{2a(-a)^n}{1+3a^2+a^3+3a} + \frac{2(n+1)\left(\frac{n}{2}+1\right)}{1+a} + \frac{(-2a-4)(n+1)}{(1+a)^2} + \frac{2}{1+3a^2+a^3+3a}$$

3. $b_n = q^n P_m(n)$ - exponential multiplied with a polinomial of degree m with respect to n

if $q \neq -a$ we have

```
> eq1_nonhom:=y(n+1)+a*y(n)=q^n*n;
```

$$eq1\_nonhom := y(n+1) + a\,y(n) = q^n\,n$$

```
> rsolve(eq1_nonhom,y(n));
```

$$y(0)(-a)^n + \frac{q(-a)^n}{(q+a)^2} - \frac{(a+2q)q^n}{q^2+2aq+a^2} + \frac{(n+1)q^n}{q+a}$$

if $q = -a$ we have

```
> eq1_nonhom:=y(n+1)+a*y(n)=(-a)^n*n;
```

$$eq1\_nonhom := y(n+1) + a\,y(n) = (-a)^n\,n$$

```
> rsolve(eq1_nonhom,y(n));
```

$$y(0)(-a)^n + \frac{2(n+1)(-a)^n}{a} - \frac{(-a)^n}{a} - \frac{(n+1)\left(\frac{n}{2}+1\right)(-a)^n}{a}$$

Applications of the first order linear difference equation

1. Interest

A sum of money that is lent to a bank earns interest. Let us denote the amount of deposit after n time periods by S_n. The interest is denoted by I_n. The relation between S_n and I_n is given by

$$S_{n+1} = S_n + I_n$$

This formula states that the amount on deposit after n periods is equal to the sum of the amount on deposit after the previous period plus the interest earned. The question now is how the interest I_n is related to S_n.

Simple Interest. A simple way of offering interest is given by the option that the interest is proportional to the initial sum S_0 deposited, called the principal,

$$I_n = p\,S_0$$

Here, p is the constant interest rate (for example example, p = 0.05 means that there is an interest rate of 5%). Using this relation it results in the following first order difference equation for S_n (n = 0, 1, ...),

$$S_{n+1} = S_n + p\,S_0$$

This equation represents a specific case of the first order difference equation where $a = -1$ and $b_n = p\,S0$.

```
> int_eq:=S(n+1)=S(n)+p*S0;
```
$$int\_eq := S(n+1) = S(n) + p\ S0$$

```
> rsolve({int_eq,S(0)=S0},S(n));
```
$$S0 - p\ S0 + p\ S0\ (n+1)$$

```
> simplify(%);
```
$$S0 + p\ S0\ n$$

```
> factor(%);
```
$$S0\ (1 + p\ n)$$

This formula is called the simple interest formula. According to this option, the amount on deposit after n periods is a linear function of the principal S_0. It is assumed that n refers to the number of years.

Exercise: We are interested to know the number k of years needed to double any initial sum S_0 in the case of simple interest.

```
> sol:=rsolve({int_eq,S(0)=S0},S(n));
```
$$sol := S0 - p\ S0 + p\ S0\ (n+1)$$

```
> s:=unapply(sol,n,S0);
```
$$s := (n, S0) \rightarrow S0 - p\ S0 + p\ S0\ (n+1)$$

```
> eq:=2*S0=s(k,S0);
```
$$eq := 2\ S0 = S0 - p\ S0 + p\ S0\ (k+1)$$

```
> solve(eq,k);
```
$$\frac{1}{p}$$

Compound Interest. Another type of interest is given if interest is added to the principal at regular intervals, which are called conversion periods. For this case, the interest formula is given by

$$I_n = \frac{p\ S_n}{r}$$

Here, r refers to the number of conversion periods per year, this means r = 12 corresponds to a conversion period of one month, and r = 4 corresponds to a conversion period of one quarter. For this interest option, the difference equation for S[n] is given by (n = 0, 1, ...),

$$S_{n+1} = S_n + \frac{p\ S_n}{r} \quad = \left(1 + \frac{p}{r}\right) S_n$$

This equation is an homogeneous first order linear difference equation, $-a = 1 + \frac{p}{r}$ and $b_n = 0$ for all n.

```
> int_eq_c:=S(n+1)=S(n)+p/r*S(n);
```
$$int\_eq\_c := S(n+1) = S(n) + \frac{p\ S(n)}{r}$$

```
> sol:=rsolve({int_eq_c,S(0)=S0},S(n));
```

$$sol := S0\left(\frac{r+p}{r}\right)^n$$

Exercise: We are interested to know the number k of years needed to double any initial sum S_0 in the case of compound interest.

```
> int_eq_c:=S(n+1)=S(n)+p/r*S(n);
```

$$int\_eq\_c := \mathrm{S}(n+1) = \mathrm{S}(n) + \frac{p\,\mathrm{S}(n)}{r}$$

```
> sol:=rsolve({int_eq_c,S(0)=S0},S(n));
```

$$sol := S0\left(\frac{r+p}{r}\right)^n$$

```
> s:=unapply(sol,n,S0);
```

$$s := (n, S0) \rightarrow S0\left(\frac{r+p}{r}\right)^n$$

```
> eq:=2*S0=s(k,S0);
```

$$eq := 2\,S0 = S0\left(\frac{r+p}{r}\right)^k$$

```
> solve(eq,k);
```

$$\frac{\ln(2)}{\ln\left(\frac{r+p}{r}\right)}$$

2. Loan Repayments

Loan repayments can be studied in a corresponding way. Let us consider the repayment of a housing loan that required any initial debt S_0 (the amount borrowed from a financial institution). The debt changes due to

(i) repayments at regular intervals made to reduce the debt, and

(ii) interest that has to be paid on the amount still owing.

By considering a compound interest, the debt S_{n+1} after $n+1$ payments is given by the following formula

$$S_{n+1} = \left(1+\frac{p}{r}\right)S_n - R$$

Here, S_{n+1} and S_n are the debt after $n+1$ and n payments, respectively, and R is the constant repayment. The term involving $\frac{p}{r}$ refers to the compound interest option.

Thus we get a nonhomogeneous first order difference equation with $-a = 1+\frac{p}{r}$ and $b_n = -R$ for all n.

```
> LR_eq:=S(n+1)= (1+p/r)*S(n)-R;
```

$$LR\_eq := \mathrm{S}(n+1) = \left(1 + \frac{p}{r}\right) \mathrm{S}(n) - R$$

```
> sol:=rsolve({LR_eq,S(0)=S0},S(n));
```

$$sol := S0 \left(\frac{r+p}{r}\right)^n + \frac{r\,R}{p} - \frac{r\,R\left(\frac{r+p}{r}\right)^n}{p}$$

The number k of conversion periods required to repay the debt can be calculated for this case on the basis of the solution formula.

```
> s:=unapply(sol,n,S0,R);
```

$$s := (n, S0, R) \rightarrow S0 \left(\frac{r+p}{r}\right)^n + \frac{r\,R}{p} - \frac{r\,R\left(\frac{r+p}{r}\right)^n}{p}$$

```
> eq:=s(k,S0,R)=0;
```

$$eq := S0 \left(\frac{r+p}{r}\right)^k + \frac{r\,R}{p} - \frac{r\,R\left(\frac{r+p}{r}\right)^k}{p} = 0$$

```
> T:=solve(eq,k);
```

$$T := \frac{\ln\left(\frac{r\,R}{-p\,S0 + r\,R}\right)}{\ln\left(\frac{r+p}{r}\right)}$$

3. The Pielou Logistic Equation (Riccati type)

$$x_{n+1} = \frac{\alpha\,x_n}{1 + \beta\,x_n}$$

```
> P_eq:=x(n+1)=alpha*x(n)/(1+beta*x(n));
```

$$P\_eq := \mathrm{x}(n+1) = \frac{\alpha\,\mathrm{x}(n)}{1 + \beta\,\mathrm{x}(n)}$$

```
> sol:=rsolve({P_eq,x(0)=x0},x(n));
```

$$sol := \mathrm{rsolve}\left(\left\{\mathrm{x}(n+1) = \frac{\alpha\,\mathrm{x}(n)}{1 + \beta\,\mathrm{x}(n)}, \mathrm{x}(0) = x0\right\}, \mathrm{x}(n)\right)$$

```
> eq1:=subs(x(n)=1/y(n),x(n+1)=1/y(n+1),P_eq);
```

$$eq1 := \frac{1}{y(n+1)} = \frac{\alpha}{y(n)\left(1+\frac{\beta}{y(n)}\right)}$$

```
> lin_eq:=y(n+1)=solve(eq1,y(n+1));
```

$$lin\_eq := y(n+1) = \frac{y(n)+\beta}{\alpha}$$

```
> sol:=rsolve({lin_eq,y(0)=1/x0},y(n));
```

$$sol := \frac{\left(\frac{1}{\alpha}\right)^n}{x0} - \frac{\beta\left(\frac{1}{\alpha}\right)^n}{-1+\alpha} + \frac{\beta}{-1+\alpha}$$

```
> xx:=unapply(1/sol,n,alpha,beta,x0);
```

$$xx := (n, \alpha, \beta, x0) \to \frac{1}{\frac{\left(\frac{1}{\alpha}\right)^n}{x0} - \frac{\beta\left(\frac{1}{\alpha}\right)^n}{-1+\alpha} + \frac{\beta}{-1+\alpha}}$$

```
> plot([[j,xx(j,3,1,0.5)]$j=0..30],style=point,symbol=circle);
```




```
> alpha:=1;
```

$$\alpha := 1$$

```
> rsolve(lin_eq,y(n));
```

$$y(0) + \beta\,(n+1) - \beta$$

```
> sol:=rsolve({lin_eq,y(0)=1/x0},y(n));
```

$$sol := \frac{1}{x0} + \beta\,(n+1) - \beta$$

```
> xx:=unapply(1/sol,n,beta,x0);
```

$$xx := (n, \beta, x0) \to \frac{1}{\frac{1}{x0} + \beta\,(n+1) - \beta}$$

```
> plot([[j,xx(j,2,1)]$j=0..100],style=point,symbol=circle);
```




```
>
```

$$\alpha := \alpha$$

Linear second order difference equation

$y_{n+2} + A\,y_{n+1} + B\,y_n = b_n$ nonhomogeneous second order difference equation

$y_{n+2} + A\,y_{n+1} + B\,y_n = 0$ homogeneous second order difference equation

General solution
$y_n = c\,y\_hom_n + y\_p_n$

where
y_hom_n is the general solution of the homogeneous equation
y_p_n is a particular solution of the nonhomogeneous equation

The case of homogeneous equation

$y_{n+2} + A\,y_{n+1} + By_n = 0$

If we want to find solution of the form $y_n = q^n$ then q must be a solution of the equation

$$q^2 + A\,q + B = 0$$

which is called the characteristic equation. The general solution of the homogeneous equation is generated by the roots of the characteristic equation.
I. If q_1 , q_2 are real distinct roots then

$$y\_hom_n = c_1\,q_1^{\ n} + c_2\,q_2^{\ n}$$

II. If $q_1 = q_2 = q$ double root then

$$y\_hom_n = (c_1 + c_2\, n)\, q^n$$

III. If $q_1 = r\,(\cos(\phi) + i\,\sin(\phi))$, $q_2 = r\,(\cos(\phi) - i\,\sin(\phi))$ are conjugated complex roots, written in the trigonometric form, then

$$y\_hom_n = r^n\,(c_1\,\cos(n\,\phi) + c_2\,\sin(n\,\phi))$$

Coefficients c_1 and c_2 can be determined from initial conditions $y_0 = \alpha$, $y_1 = \beta$

The case of nonhomogeneous equation

General solution

$$y_n = y\_hom_n + y\_p_n$$

y_hom_n is the general solution of the homogeneous equation

y_p_n, the particular solution of the nonhomogeneous equation, generaly, can be found only in some pariticular cases of b_n as in the case of the first order linear difference equation

Examples:

a) $y_{n+2} - 5\,y_{n+1} + 6\,y_n = 0$

b) $y_{n+2} - 4\,y_{n+1} + 4\,y_n = 0$

c) $y_{n+2} - 5\,y_{n+1} + 6\,y_n = 4\,n - 2$

d) $y_{n+2} - 5\,y_{n+1} + 6\,y_n = 4^n\,(a\,n + b)$

```
> eq_a:=y(n+2)-5*y(n+1)+6*y(n)=0;
```

$$eq\_a := y(n+2) - 5\,y(n+1) + 6\,y(n) = 0$$

```
> ch_eq:=q^2-5*q+6=0;
```

$$ch\_eq := q^2 - 5\,q + 6 = 0$$

```
> solve(ch_eq,q);
```

$$3,\ 2$$

```
> rsolve(eq_a,y(n));
```

$$-(2\,y(0) - y(1))\,3^n - (-3\,y(0) + y(1))\,2^n$$

```
> eq_b:=y(n+2)-4*y(n+1)+4*y(n)=0;
```

$$eq\_b := y(n+2) - 4\,y(n+1) + 4\,y(n) = 0$$

```
> ch_eq:=q^2-4*q+4=0;
```

$$ch\_eq := q^2 - 4\,q + 4 = 0$$

```
> solve(ch_eq,q);
```

$$2,\ 2$$

```
> rsolve(eq_b,y(n));
```

$$\left(-y(0)+\frac{1}{2}y(1)\right)(n+1)\,2^n-\left(\frac{1}{2}y(1)-2\,y(0)\right)2^n$$

```
> eq_c:=y(n+2)-5*y(n+1)+6*y(n)=4*n-2;
```

$$eq\_c := y(n+2)-5\,y(n+1)+6\,y(n)=4\,n-2$$

```
> sol:=rsolve(eq_c,y(n));
```

$$sol := -(2\,y(0)-y(1))\,3^n-(-3\,y(0)+y(1))\,2^n-2\,2^n+2\,n+2$$

```
> sol:=rsolve({eq_c,y(0)=0,y(1)=1},y(n));
```

$$sol := 3^n-3\,2^n+2\,n+2$$

```
> eq_d:=y(n+2)-5*y(n+1)+6*y(n)=4^n*(a*n+b);
```

$$eq\_d := y(n+2)-5\,y(n+1)+6\,y(n)=4^n\,(a\,n+b)$$

```
> sol:=rsolve(eq_d,y(n));
```

$$sol := -(2\,y(0)-y(1))\,3^n-(-3\,y(0)+y(1))\,2^n+\frac{a\,(n+1)\,4^n}{2}-(-4\,a+b)\,3^n-\left(\frac{7\,a}{2}-\frac{b}{2}\right)4^n-\left(a-\frac{b}{2}\right)2^n$$

```
> a:=2;b:=4;
```

$$a := 2$$

$$b := 4$$

```
> sol:=rsolve({eq_c,y(0)=1,y(1)=0},y(n));
```

$$sol := 3\,2^n+2\,3^n+(n+1)\,4^n-5\,4^n$$