

Solving differential equations with MAPLE

Functions and graphic representation

A single variable function can be defined as follows:

```
> f:=x->sin(x)/x;
```

$$f:=x \rightarrow \frac{\sin(x)}{x}$$

```
> f(3*Pi/2),f(1.5);
```

$$-\frac{2}{3\pi}, 0.6649966577$$


```
> f(a+b);
```

$$\frac{\sin(a+b)}{a+b}$$

For graphical representation we need to load **plots** package using **with** command


```
> with(plots):
```

```
> plot(f(x),x=-1..1);
```




We can represent more than one function in the same window:

```
> plot([f(x),f(2*x),sin(x)],x=0..2*Pi,color=[red,blue,green]);
```




In the case of discontinuous points we need to use the option **discont = true**:

```
> plot(tan(x), x = -2*Pi..2*Pi, y = -4..4, discont = true);
```




If a curve is given in a parametric form (for example: $x(t) = \sin(t)$, $y(t) = \cos(t)$, $t = 0 \dots \pi$) we use the instruction:

```
> plot([sin(t),cos(t),t=0..3/2*Pi]);
```




In the case of a curve given by the implicit equation we use the instruction **implicitplot**:

```
> implicitplot(x^2+y^2=1,x=-10..10,y=-10..10,numpoints=10000);
```




In the case we need to visualize the parameter dependence of a function we can use the command **animate** (right click on the image, select *Animation* and *Play*)

```
> animate(sin(x*t),x=-4*Pi..4*Pi,t=0..2,color=red);
```



If we need more precision we can increase the number of point and frames:

```
> animate(sin(x*t),x=-4*Pi..4*Pi,t=0..2,color=red,numpoints=
100,frames=100);
```




A function with more than one variable can be defined as follows:

```
> g:=(x,y)->sin(x)*exp(-y);
```

$$g := (x, y) \rightarrow \sin(x) e^{-y}$$


For 3d graphical representation we use the command **plot3d**:

```
> plot3d(g(x,y),x=0..Pi,y=0..3,axes=boxed);
```



The animation of 3D graphs can be made using the instruction **animate3d**:

```
> animate3d(g(t*x,y),x=0..Pi,y=0..3,t=0..2);
```

The derivation of the functions

The derivation of the functions can be made in two ways: using **diff** command or using the derivation operator **D**:

```
> f:=x->exp(x^2)+3;
```

$$f:=x \rightarrow e^{x^2} + 3$$

The **diff** command execute the derivation of the given expresion with respect to the specified variable. The derivation operator **D** returns the derivate as a function.

```
> diff(f(x),x);
```

$$2 x e^{x^2}$$

```
> diff(f(x),x)
```

$$2 x e^{x^2}$$

```
> eval(diff(f(x),x),x=1)
```

$$2 e$$

the second order derivate is given by

```
> diff(f(x),x,x);
```

$$2 e^{x^2} + 4 x^2 e^{x^2}$$

also we can use the option **t\$n** to get n-order derivative

```
> diff(f(x),x$2);
```

$$2 e^{x^2} + 4 x^2 e^{x^2}$$

```
> diff(f(x),x$3);
```

$$12 x e^{x^2} + 8 x^3 e^{x^2}$$

```
> diff(x^2+y^2,y);
```

$$2 y$$

Using the derivation operator:

```
> D(f)(x);
```

$$2 x e^{x^2}$$

```
> D(f)(1);
```

$$2 e$$

```
> (D@D)(f)(x);
```

$$2 e^{x^2} + 4 x^2 e^{x^2}$$

```
> (D@D)(f)(1);
```

$$6 e$$

```
> (D@@2)(f)(x);
```

$$2 e^{x^2} + 4 x^2 e^{x^2}$$

```
> (D@D@D)(f)(x);
```

$$12 x e^{x^2} + 8 x^3 e^{x^2}$$

```
> (D@@10)(f)(x);
```

$$30240 e^{x^2} + 302400 x^2 e^{x^2} + 403200 x^4 e^{x^2} + 161280 x^6 e^{x^2} + 23040 x^8 e^{x^2} + 1024 x^{10} e^{x^2}$$

Initialization of the solving ODE package

```
> restart:
```

clears the memory of all previously saved values and variables

```
> with(DEtools):
```

load the differential equations package

```
> with(plots):
```

load the graphical package

Define and solve a first order differential equation

Let consider the differential equation $\frac{d}{dx} y(x) = k y(x)$ where k is a real coefficient. The differential equation can be introduced in MAPLE as follows:

```
> diff_eq1:=diff(y(x),x) = k*y(x);
```

$$\text{diff_eq1} := \frac{d}{dx} y(x) = k y(x)$$

To obtain the general solution of the equation use **dsolve** command

```
> dsolve(diff_eq1,y(x));
```

$$y(x) = _CI e^{kx}$$

The general solution is seen as an expression. Notice that the undetermined constant is called $_CI$
How can we manipulate this expression?

We can use the function definition command:

```
> sol:=(x,k,c)->c*exp(k*x);
```

$$sol := (x, k, c) \rightarrow c e^{kx}$$

If the expression of the solution is too complicated we can use the command **rhs** (*right hand side*) and **unapply** in order to obtain the solution as a function

```
> right_hand_expr:=rhs(dsolve(diff_eq1,y(x)));
```

$$right_hand_expr := _CI e^{kx}$$

Using the **unapply** command we transform the expression sol1 into a function specifying the variables:

```
> sol1:=unapply(right_hand_expr,x,k,\_CI);
```

$$sol1 := (x, k, _CI) \rightarrow _CI e^{kx}$$

and we get the same result.

▼ The graphics of ODE solutions


Let suppose that $k := 2$. Then the corresponding general solution is:

```
> y:=(x,c)->sol(x,2,c);
```


$$y := (x, c) \rightarrow sol(x, 2, c)$$

To draw the solutions curves you just assign some values for the constant c. For example take $c:=1$ $c:=2$ and $c:=-1$

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2);
```




```
> animate( y(x,c),x=-2..2,c=-2..2,frames=50);
```




If you want to obtain the solutions with some specified colors use the command:

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2,color=[black,red,blue]);
```



Also you specify the window of the graphic:

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2,y=-10..10,color=[black,  
red,blue]);
```




Using this way of manipulation for the solution you can see also how the solution depends on the k parameter. Let us consider $c:=1$ and assign some values for the parameter k .

```
> y1:=(x,k)->sol(x,k,1);
```

```
y1 := (x, k) → sol(x, k, 1)
```

```
> plot([y1(x,0.2),y(x,0.5),y(x,-1)],x=-2..2,y=-10..10,color=[black,red,blue]);
```



Solving an IVP

Suppose that we want to solve the IVP $\frac{d}{dx} y(x) = k y(x)$ with the initial condition $y(0) = 1$

```
> restart:with(DEtools):
```

```
> diff_eq:=diff(y(x),x) = k*y(x);
```

$$\text{diff_eq} := \frac{d}{dx} y(x) = k y(x)$$

```
> in_cond:=y(0)=1;
```

$$\text{in_cond} := y(0) = 1$$

```
> dsolve({diff_eq,in_cond},y(x));
```

$$y(x) = e^{kx}$$

Let consider the case $k = 2$

```
> k:=2;
```

$$k := 2$$


```
> sol:=dsolve({diff_eq,in_cond},y(x));
```

$$\text{sol} := y(x) = e^{2x}$$

```
> yy:=unapply(rhs(sol),x);
```




```
yy := x → e2x  
> plot(yy(x), x=-1..1, y=0..4);
```




You can obtain the graph the IVP directly using the command DEplot:

```
> DEplot(diff_eq, y(x), x=-1..1, [[in_cond]]);
```



In this graph is also represented the direction field of the equation. If you want the graphs of the solutions for different initial condition ($y(0) = 1, y(0) = 1.5, y(0) = 2$) you can use the same command and specify the list of initial conditions:

```
> DEplot(diff_eq,y(x),x=-1..1,[y(0)=1],[y(0)=1.5],[y(0)=2]));
```



First Order Solvable Differential Equations

Separable Differential Equations

$$\frac{d}{dx} y(x) = f(x) g(y(x))$$

```
> restart;with(DEtools):
> sep_eq:=diff(y(x),x)=f(x)*g(y(x));
      sep_eq :=  $\frac{d}{dx} y(x) = f(x) g(y(x))$ 
```

```
> dsolve(sep_eq,y(x));
```

$$\int f(x) \, dx - \left(\int \frac{1}{g(a)} \, da \right) + _C1 = 0$$

Examples:

a) $y' = 2x(1 + y^2)$

b) $(x^2 - 1) y' + 2 x y^2 = 0$

c) $y' = e^{x+y}$

```
> eq_a:=diff(y(x),x)=2*x*(1+(y(x))^2);
```

$$eq_a := \frac{d}{dx} y(x) = 2 x (1 + y(x)^2)$$

```
> dsolve(eq_a,y(x));
```

$$y(x) = \tan(x^2 + 2 _CI)$$

```
> eq_b:=(x^2-1)*diff(y(x),x)+2*x*(y(x))^2=0;
```

$$eq_b := (x^2 - 1) \left(\frac{d}{dx} y(x) \right) + 2 y(x)^2 x = 0$$

```
> dsolve(eq_b,y(x));
```

$$y(x) = \frac{1}{\ln(x-1) + \ln(x+1) + _CI}$$

```
> eq_c:=diff(y(x),x)=exp(x+y(x));
```

$$eq_c := \frac{d}{dx} y(x) = e^{x+y(x)}$$

```
> dsolve(eq_c,y(x));
```

$$y(x) = \ln\left(-\frac{1}{e^x + _CI}\right)$$

Homogeneous (in the Euler sense) Differential Equations

$$\frac{d}{dx} y(x) = F\left(\frac{y(x)}{x}\right)$$

```
> hom_eq:=diff(y(x),x)=F(y(x)/x);
```

$$hom_eq := \frac{d}{dx} y(x) = F\left(\frac{y(x)}{x}\right)$$

```
> dsolve(hom_eq,y(x));
```

$$y(x) = \text{RootOf}\left(-\left(\int \frac{1}{F(_a) - _a} d_a\right) + \ln(x) + _CI\right) x$$

Examples:

a) $2 x^2 y' = x^2 + y^2$

b) $x y' = \sqrt{x^2 - y^2} + y$

c) $2 x^3 y' = y^3 + x^2 y$

```
> eq_a:=2*x^2*diff(y(x),x)=x^2+y(x)^2;
```

$$eq_a := 2 \left(\frac{d}{dx} y(x) \right) x^2 = x^2 + y(x)^2$$

```
> dsolve(eq_a,y(x));
```

$$y(x) = \frac{x (\ln(x) + _CI - 2)}{\ln(x) + _CI}$$

```
> eq_b:=x*diff(y(x),x)=sqrt(x^2-y(x)^2)+y(x);
```

$$eq_b := x \left(\frac{d}{dx} y(x) \right) = \sqrt{x^2 - y(x)^2} + y(x)$$

```
> ans:=dsolve(eq_b,y(x));
```

$$ans := -\arctan\left(\frac{y(x)}{\sqrt{x^2 - y(x)^2}}\right) + \ln(x) - _CI = 0$$

```
> solve(ans,y(x));
```

$$-\tan(-\ln(x) + _CI) \sqrt{\frac{x^2}{\tan(-\ln(x) + _CI)^2 + 1}}$$

```
> eq_c:=2*x^3*diff(y(x),x)=y(x)^3+x^2*y(x);
```

$$eq_c := 2x^3 \left(\frac{d}{dx} y(x) \right) = y(x)^3 + y(x)x^2$$

```
> dsolve(eq_c,y(x));
```

$$y(x) = \frac{x}{\sqrt{_CI x + 1}}, y(x) = -\frac{x}{\sqrt{_CI x + 1}}$$

First Order Linear Differential Equations

$$\frac{d}{dx} y(x) + P(x) y(x) = Q(x)$$

```
> lin_eq:=diff(y(x),x)+P(x)*y(x)=Q(x);
```

$$lin_eq := \frac{d}{dx} y(x) + P(x) y(x) = Q(x)$$

```
> dsolve(lin_eq,y(x));
```

$$y(x) = \left(\int Q(x) e^{\int P(x) dx} dx + _CI \right) e^{\int (-P(x)) dx}$$

Examples:

a) $y' + y \tan(x) = \frac{1}{\cos(x)}$

b) $y' + \frac{y(x)}{x} = 3x$

c) $xy' + y = e^x$

```
> eq_a:=diff(y(x),x)+tan(x)*y(x)=1/cos(x);
```

$$eq_a := \frac{d}{dx} y(x) + \tan(x) y(x) = \frac{1}{\cos(x)}$$

```
> dsolve(eq_a,y(x));
```

$$y(x) = \cos(x) \tan(x) + \cos(x) _CI$$

```
> eq_b:=diff(y(x),x)+1/x*y(x)=3*x;
```

$$eq_b := \frac{d}{dx} y(x) + \frac{y(x)}{x} = 3x$$

```
> dsolve(eq_b,y(x));
```

$$y(x) = \frac{x^3 + C1}{x}$$

```
> eq_c:=x*diff(y(x),x)+y(x)=exp(x);
```

$$eq_c := x \left(\frac{d}{dx} y(x) \right) + y(x) = e^x$$

```
> dsolve(eq_c,y(x));
```

$$y(x) = \frac{e^x + C1}{x}$$

Solving a second order ODE

```
> restart;
```

```
> with(DEtools):
```

```
> with(plots):
```

Consider the linear differential equation with the constant coefficients $y'' + 3y' + 2y = 1 + x^2$

```
> deq1:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=1+x^2;
```

$$deq1 := \frac{d^2}{dx^2} y(x) + 3 \left(\frac{d}{dx} y(x) \right) + 2 y(x) = x^2 + 1$$

To obtain the general solution we use the dsolve command

```
> dsolve(deq1,y(x));
```

$$y(x) = \frac{1}{2} x^2 - \frac{3}{2} x + \frac{9}{4} - e^{-2x} C1 + e^{-x} C2$$

If we want to study the solution we can use the same technique as in the previous section in order to draw the solution graph.

```
> sol:=dsolve(deq1,y(x));
```

$$sol := y(x) = \frac{1}{2} x^2 - \frac{3}{2} x + \frac{9}{4} - e^{-2x} C1 + e^{-x} C2$$

```
> right_hand:=rhs(sol);
```


$$right_hand := \frac{1}{2} x^2 - \frac{3}{2} x + \frac{9}{4} - e^{-2x} C1 + e^{-x} C2$$

```
> y_sol:=unapply(right_hand,x,_C1,_C2);
```

$$y_sol := (x, C1, C2) \rightarrow \frac{1}{2} x^2 - \frac{3}{2} x + \frac{9}{4} - e^{-2x} C1 + e^{-x} C2$$

Now we are able to one ore more than one solution graphs using the **plot** command.

```
> plot([y_sol(x,0,0),y_sol(x,0,1),y_sol(x,1,0)],x=-2..2,y=-10..10);
```



In the case of initial value problem we have two initial conditions, for example lets take $y(0) = 1$ and $y'(0) = 0$.

```
> in_cond:=y(0)=1,D(y)(0)=0;
```

```
in_cond:=y(0)=1,D(y)(0)=0
```


To obtain the corresponding solution we use **dsolve** command in the following form:

```
> dsolve({deq1,in_cond},y(x));
```

$$y(x) = \frac{1}{2} x^2 - \frac{3}{2} x + \frac{9}{4} - \frac{1}{4} e^{-2x} - e^{-x}$$


Now we can use the previous technique (**rhs** and **unapply** commands) to construct the solution as a function and after that to represent its graph or we can obtain this graph directly using **DEplot** command.

```
> DEplot(deq1,y(x),x=-2..2,y=-10..10,[[in_cond]]);
```



If we need to draw more than one solution corresponding to different initial value problem we can use the same **DEplot** command specifying the list of initial conditions:

```
> DEplot(deq1,y(x),x=-2..2,[[y(0)=1,D(y)(0)=0],[y(0)=1,D(y)(0)=1],[y(0)=0,D(y)(0)=0]],y=-10..10,linecolor=red);
```

The general second order linear DE, $y'' + P(x)y' + Q(x)y = f(x)$:

Note: Maple is unable to solve most second-order DE's explicitly. For information on numerically solving DE's, see Numerical Solutions with dsolve.

Consider the differential equation $y'' + x y'(x) + y(x) = \sin(x)$. Try to use the **dsolve** command.

```
> deq2:=diff(y(x),x$2)+x*diff(y(x),x)+y(x)=sin(x);
```

$$deq2 := \frac{d^2}{dx^2} y(x) + x \left(\frac{d}{dx} y(x) \right) + y(x) = \sin(x)$$

```
> dsolve(deq2,y(x));
```

$$y(x) = e^{-\frac{1}{2}x^2} \operatorname{erf}\left(\frac{1}{2} \sqrt{2} x\right) C_1 + e^{-\frac{1}{2}x^2} C_2 + \frac{1}{4} \sqrt{\pi} e^{\frac{1}{2}x^2} \left(\operatorname{erf}\left(\frac{1}{2} \sqrt{2} x - \frac{1}{2} \sqrt{2}\right) + \operatorname{erf}\left(\frac{1}{2} \sqrt{2} x + \frac{1}{2} \sqrt{2}\right) \right) e^{-\frac{1}{2}x^2}$$

```
> ?erf
```

```
> in_cond2:=y(0)=1,D(y)(0)=1;
```

$$in_cond2 := y(0) = 1, D(y)(0) = 1$$

```
> dsolve({deq2,in_cond2},y(x));
```

$$y(x) = -\frac{1}{2} I e^{-\frac{1}{2} x^2} \operatorname{erf}\left(\frac{1}{2} I \sqrt{2} x\right) \sqrt{\pi} \left(e^{\frac{1}{2}} e^{-\frac{1}{2}} + 1\right) \sqrt{2} + e^{-\frac{1}{2} x^2} + \frac{1}{4} I \sqrt{\pi} e^{\frac{1}{2}} \sqrt{2} \left(\operatorname{erf}\left(\frac{1}{2} I \sqrt{2} x - \frac{1}{2} \sqrt{2}\right) + \operatorname{erf}\left(\frac{1}{2} I \sqrt{2} x + \frac{1}{2} \sqrt{2}\right)\right) e^{-\frac{1}{2} x^2}$$

Maple expresses the solution in terms of a modified Bessel function I and the error function **erf**.

We can obtain the numerical solution using in the `dsolve` command the option '**type=numeric**' and the `odeplot` comand to draw the corresponding graph.

```
> n_sol:=dsolve({deq2,in_cond2},y(x),type=numeric):
> odeplot(n_sol);
```

