Mathematical models given by difference equations

Discrete mathematical models for single population

In modeling the dynamics of a more complex population, such as populations of insects, fish, or mammals, it is necessary to take into account the birth and death factors that may occur during the time period in which the growth of the respective population is studied.
To simplify the mathematical modeling, we will assume that we are studying the dynamics of a large population. Thus, we can treat the population as a whole and assume that its development is governed by the average behavior of its members.

We will also make the following assumptions:

- Each member of the population contributes equally to demographic growth, meaning they produce the same number of new members;
- Each member has the same chance of survival or death;
- Age differences between members of the population are neglected;
- We consider an isolated population, meaning the phenomenon of migration is neglected.

x_n - the size of the population at time n or after n seasons considered
x_0 - the initial size of the population

$\frac{x_{n+1} - x_n}{x_n}$ - per capita growth rate of the population.

The Unlimited Growth Model

In this model, the per capita growth rate is given by the difference between the per capita birth rate and the per capita death rate, these rates are considered constant in time.

α - per capita birth rate
β - per capita death rate

$r = \alpha - \beta = \text{constant}$

$$\frac{x_{n+1} - x_n}{x_n} = \alpha - \beta$$

$$x_{n+1} = x_n + (\alpha - \beta)x_n$$

$$x_{n+1} = (1 + r)x_n$$

```
> eq:=x(n+1)=(1+r)*x(n);
```

$$eq := x(n+1) = (1+r)\,x(n)$$

```
> rsolve(eq,x(n));
```

$$x(0)\,(1+r)^n$$

We have the following conclusions:

If $\alpha < \beta$,i.e. $r < 0$, then $\lim_{n\to\infty} x_n = 0$ - the population disappears in time;
If $\alpha = \beta$, i.e. $r = 0$ then $x_n \equiv x_0$ - the population remains constant in time;
If $\alpha > \beta$, i.e., $0 < r$, then $\lim_{n\to\infty} x_n = +\infty$ - the population grows without limit.

The Discrete Logistic Model

Studies have shown that when the population grows, the death rate increases, while the birth rate decreases. This is due to phenomena such as overpopulation and competition for resources.
The population develops in an environment that provides the necessary resources for survival.
The carrying capacity of an environment represents the maximum number of individuals that it can sustain, or the population size at which the birth rate equals the death rate. We denote this environmental carrying capacity by K.
To model overpopulation and competition for resources, we will consider in the unlimited growth model that the per capita growth rate, denoted by r, is variable and depends on the number of individuals at that moment, so

$$r = r(x_n)$$

The function $r(x_n)$ must satisfy the following conditions:

- Due to overpopulation, the per capita death rate must increase while the birth rate must decrease; in other words, the function $r(x_n)$ must be a decreasing function.
- If $x_n \to K$, then $r(x_n) \to 0$, meaning that if the population tends toward the environment's carrying capacity, then the per capita growth rate must tend toward zero.
- If $x_n \to 0$ then $r(x_n) \to r_0$, meaning that if the population is small, the effects of overpopulation have a small influence on its development, and the population grows according to the unlimited growth model with a constant per capita growth rate r_0, called *the unrestricted growth rate.*

There are several ways to choose the function $r(x_n)$ so that the above conditions are met. The simplest choice is given by the linear interpolation of the points $(K, 0)$ and $(0, r_0)$ (Verhulst, 1838)

$$r(x) = r_0 \left(1 - \frac{x}{K}\right)$$

so, the logistic model is

$$x_{n+1} = x_n + x_n \cdot r_0 \left(1 - \frac{x_n}{K}\right)$$

```
> restart;
> f:=x->x+x*r0*(1-x/K);
```

$$f := x \to x + x\, r0 \left(1 - \frac{x}{K}\right)$$

```
> eq:=x=f(x);
```

$$eq := x = x + x\, r0 \left(1 - \frac{x}{K}\right)$$

```
> eqp:=solve(eq,x);
```

$$eqp := 0, K$$

```
> D(f)(0);
```

$$1 + r0$$

if $|1 + r0| < 1$ implies $-2 < r0 < 0$ so, in these conditions $x=0$ is locally asymptotically stable
if $|1 + r0| > 1$ implies $r0 < -2$ or $r0 > 0$, in these conditions $x=0$ is unstable

```
> D(f)(K);
```

$$1 - r0$$

if $|1 - r0| < 1$ implies $0 < r0 < 2$ so, in these conditions $x=K$ is locally asymptotically stable.
if $|1 - r0| > 1$ implies $r0 < 0$ or $r0 > 2$, in these conditions $x=K$ is unstable.

```
> eqd:=x(n+1)=f(x(n));
```

$$eqd := x(n+1) = x(n) + x(n)\, r0 \left(1 - \frac{x(n)}{K}\right)$$

```
> rsolve(eqd,x(n));
```

$$rsolve\left(x(n+1) = x(n) + x(n)\, r0 \left(1 - \frac{x(n)}{K}\right), x(n)\right)$$

Time in hours	Observed yeast biomass
0	9.6
1	18.3
2	29
3	47.2
4	71.1
5	119.1
6	174.6
7	257.3
8	350.7
9	441
10	513.3
11	559.7
12	594.8
13	629.4
14	640.8
15	651.1
16	655.9
17	659.6
18	661.8

```
> restart;
```

```
> p:=[9.6,18.3,29,47.2,71.1,119.1,174.6,257.3,350.7,441,513.3,
  559.7,594.8,629.4,640.8,651.1,655.9,659.6,661.8];
```

$p := [9.6, 18.3, 29, 47.2, 71.1, 119.1, 174.6, 257.3, 350.7, 441, 513.3, 559.7, 594.8, 629.4, 640.8, 651.1, 655.9, 659.6, 661.8]$

```
real_data:=[[n,p[n]]$n=1..19]:
> plot(real_data,style=point,symbol=circle);
```




```
> for i from 1 to 18 do
    delta_p[i]:=p[i+1]-p[i]
  end do;
```

$$delta\_p_1 := 8.7$$

$$delta\_p_2 := 10.7$$

$$delta\_p_3 := 18.2$$

$$delta\_p_4 := 23.9$$

$$delta\_p_5 := 48.0$$

$$delta\_p_6 := 55.5$$

$$delta\_p_7 := 82.7$$

$$delta\_p_8 := 93.4$$

$$delta\_p_9 := 90.3$$

$$delta\_p_{10} := 72.3$$

$$delta\_p_{11} := 46.4$$

$$delta\_p_{12} := 35.1$$

$$delta\_p_{13} := 34.6$$

$$delta\_p_{14} := 11.4$$

$$delta\_p_{15} := 10.3$$

$$delta\_p_{16} := 4.8$$

$$delta\_p_{17} := 3.7$$

$$delta\_p_{18} := 2.2$$

From the graph of population versus time, the population appears to be approaching to a limiting value (carrying capacity). Based on our graph we estimate that the carrying capacity is 670. As p_n approaches to 670 the change slows considerable, this shows us that the change depends on $670 - p_n$ and, also, on the biomass p_n. Lets plot the change of biomass Δp_n versus $(670 - p_n)\, p_n$

```
> plot([[(670-p[n])*p[n],delta_p[n]]$n=1..18],style=point,
  symbol=circle,labels=["(670-p[n])*p[n]", Change_of_biomass],
  labeldirections=[HORIZONTAL, VERTICAL]);
```



It seems to have some proportionality here, so

$\Delta p_n = r_0 (670 - p_n) p_n$

Now, we find the value of *k* using two data (11th and 13th point of the graph)

```
> r0:=(delta_p[13]-delta_p[11])/((670-p[13])*p[13]-(670-p[11])*
  p[11]);
```

$r0 := 0.0003304845379$

```
> p_est[1]:=p[1];
```

$p\_est_1 := 9.6$

```
> for i from 1 to 19 do
    p_est[i+1]:=p[i]+r0*(670-p_est[i])*p_est[i]
  end do:
> est_data:=[[n,p_est[n]]$n=1..18]:
> plot([real_data,est_data],style=point,symbol=[circle,cross]);
```



Other choices for $r(x)$

For

$$r(x) = (r_0 + 1) \cdot e^{1 - \frac{x}{K}} - 1$$

Ricker model (1954): $x_{n+1} = x_n \cdot e^{a \cdot \left(1 - \frac{x}{K}\right)}$ where $a = \ln(r_0 + 1)$

For

$$r(x) = \frac{r_0(K - x) + 1}{r_0 x + K}$$

Verhulst model: $x_{n+1} = \frac{a x_n}{x_n + A}$ where $a = \frac{(r_0 + 1) K}{r_0}$ and $A = \frac{K}{r_0}$

Pielou model: $x_{n+1} = \dfrac{\alpha x_n}{1 + \beta x_n}$ where $\alpha = r_0 + 1$ and $\beta = \dfrac{K}{r_0}$

Propagation of Annual Plants Model

The material of this section comes from Edelstein–Keshet of plant propagation. Our objective here is to develop a mathematical model that describes the number of plants in any desired generation. It is known that plants produce seeds at the end of their growth season (say August), after which they die. Furthermore, only a fraction of these seeds survive the winter, and those that survive germinate at the beginning of the season (say May), giving rise to a new generation of plants.

Let
γ = number of seeds produced per plant in August,
α = fraction of one-year-old seeds that germinate in May,
β = fraction of two-year-old seeds that germinate in May,
σ = fraction of seeds that survive a given winter.
If p_n denotes the number of plants in generation n, then

$$p_n = \alpha\, s1_n + \beta\, s2_n$$

where
$s1_n$ = the number of one-year-old seeds in April
$s2_n$ = the number of two-year-old seeds in April
Observe that the number of seeds left after germination may be written as
seeds left = (fraction of not germinated) x (original number of seeds in April)
thus

$$ss1_n = (1 - \alpha)\, s1_n$$
$$ss2_n = (1 - \beta)\, s2_n$$

where
$ss1_n$ = the number of one-year-old not germinated seeds left in May
$ss2_n$ = the number of two-year-old not germinated seeds left in May
New seeds $s0_n$ (0-year-old) are produced in August at the rate of γ per plant,

$$s0_n = \gamma p_n$$

After winter, seeds $s0_n$ that were new in generation n will be one year old in the next generation $n+1$, and a fraction $\sigma\, s0_n$ of them will survive. Hence

$$s1_{n+1} = \sigma\, s0_n$$

or

$$s1_{n+1} = \sigma\, \gamma p_n$$

Simlarly,

$$s2_n = \sigma\, ss1_n$$

which yields

$$s2_{n+1} = \sigma\, (1 - \alpha)\, s1_n$$
$$s2_{n+1} = \sigma^2\, (1 - \alpha)\, \gamma p_{n-1}$$

Substituting for $s1_n$, $s2_n$ in expressions of p_n gives

$$p_n = \alpha\, s1_n + \beta\, s2_n$$

$$p_n = \alpha\,\sigma\,\gamma p_{n-1} + \beta\,\sigma^2\,(1-\alpha)\,\gamma p_{n-2}$$

or

$$p_{n+2} = \alpha\,\sigma\,\gamma p_{n+1} + \beta\,\sigma^2\,(1-\alpha)\,\gamma p_n$$

```
> restart;
> char_eq:=q^2-alpha*sigma*gamma*q-beta*sigma^2*(1-alpha)*
  gamma=0;
```

$$char\_eq := q^2 - \alpha\,\sigma\,\gamma\,q - \beta\,\sigma^2\,(1-\alpha)\,\gamma = 0$$

```
> solve(char_eq,q);
```

$$\left(\frac{1}{2}\,\gamma\alpha + \frac{1}{2}\,\sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\right)\sigma,\ \left(\frac{1}{2}\,\gamma\alpha - \frac{1}{2}\,\sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\right)\sigma$$

```
> eq:=p(n+2)=alpha*sigma*gamma*p(n+1)+beta*sigma^2*(1-alpha)*
  gamma*p(n);
```

$$eq := p(n+2) = \alpha\,\sigma\,\gamma p(n+1) + \beta\,\sigma^2\,(1-\alpha)\,\gamma p(n)$$

```
> rsolve(eq,p(n));
```

$$\left(\left(\gamma^2 p(0)\,\sigma\,\alpha^2 - \gamma\sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\;p(0)\,\alpha\,\sigma - 2\,\gamma p(0)\,\alpha\,\beta\,\sigma + 2\,\gamma p(0)\,\beta\,\sigma\right.\right.$$
$$\left.- \gamma p(1)\,\alpha + \sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\;p(1)\right)\left(-\frac{2\,\alpha\,\beta\,\gamma\sigma - 2\,\beta\,\gamma\sigma}{-\gamma\alpha + \sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}}\right)^n\Bigg)$$
$$\Bigg/\left(\sigma\sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\,\left(-\gamma\alpha + \sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\right)\right) + \Bigg(\Bigg($$
$$-\gamma^2 p(0)\,\sigma\,\alpha^2 - \gamma\sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\;p(0)\,\alpha\,\sigma + 2\,\gamma p(0)\,\alpha\,\beta\,\sigma - 2\,\gamma p(0)\,\beta\,\sigma$$
$$\left.+ \gamma p(1)\,\alpha + \sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\;p(1)\right)\left(-\frac{2\,\alpha\,\beta\,\gamma\sigma - 2\,\beta\,\gamma\sigma}{-\gamma\alpha - \sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}}\right)^n\Bigg)$$
$$\Bigg/\left(\sigma\sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\,\left(-\gamma\alpha - \sqrt{\alpha^2\gamma^2 - 4\,\alpha\,\beta\,\gamma + 4\,\beta\,\gamma}\right)\right)$$

```
> char_eq:=q^2=alpha*sigma*gamma*q+beta*sigma^2*(1-alpha)*
  gamma;
```

$$char\_eq := q^2 = \alpha\,\sigma\,\gamma\,q + \beta\,\sigma^2\,(1-\alpha)\,\gamma$$

```
> p[1]:=alpha*sigma*gamma;
```

$$p_1 := \gamma\,\alpha\,\sigma$$

```
> p[2]:=beta*sigma^2*(1-alpha)*gamma;
```

$$p_2 := \beta\,\sigma^2\,(1-\alpha)\,\gamma$$

The Schur-Cohn Criterion
For the equation the equation

$$q^2 = p_1 q + p_2$$

we have:
$|q_{1,2}| < 1$ if and only if $|p_1| < 1 - p_2 < 2$

```
> p[1]<1-p[2];
```

$$\gamma \alpha \sigma < 1 - \beta \sigma^2 (1-\alpha) \gamma$$

```
> 1-p[2]<2;
```

$$-\beta \sigma^2 (1-\alpha) \gamma < 1$$

The Pielou delay logistic model

In the Pielou delay model, the population at the next time step depends on the population some time ago (one season before), not just the current population or we can consider the per capita growth rate depends on the population size from the previous season, not the current one.

$$r = r(x_{n-1}) = \frac{\alpha}{1 + \beta x_{n-1}}$$

$$x_{n+1} = \frac{\alpha x_n}{1 + \beta x_{n-1}}$$

```
> restart;
> f:=(u,v)->alpha*u/(1+beta*v);
```

$$f := (u, v) \rightarrow \frac{\alpha u}{1 + \beta v}$$

```
> eq:=x=f(x,x);
```

$$eq := x = \frac{\alpha x}{\beta x + 1}$$

```
> eqp:=solve(eq,x);
```

$$eqp := 0, \frac{\alpha - 1}{\beta}$$

```
> p1:=D[1](f)(0,0);
```

$$p1 := \alpha$$

```
> p2:=D[2](f)(0,0);
```

$$p2 := 0$$

```
> lineq:=y(n+1)=p1*y(n)+p2*y(n-1);
```

$$lineq := y(n+1) = \alpha\, y(n)$$

```
> chareq:=q^2=p1*q+p2;
```

$$chareq := q^2 = \alpha\, q$$

```
> rr:=solve(chareq,q);
```

$$rr := 0, \alpha$$

```
> x1:=(alpha-1)/beta;
```

$$x1 := \frac{\alpha - 1}{\beta}$$

```
> p1:=D[1](f)(x1,x1);
```

$$p1 := 1$$

```
> p2:=D[2](f)(x1,x1);
```

$$p2 := -\frac{\alpha - 1}{\alpha}$$

```
> chareq:=q^2=p1*q+p2;
```

$$chareq := q^2 = q - \frac{\alpha - 1}{\alpha}$$

```
> rr:=solve(chareq,q);
```

$$rr := \frac{1}{2}\,\frac{\alpha + \sqrt{-3\,\alpha^2 + 4\,\alpha}}{\alpha},\ -\frac{1}{2}\,\frac{-\alpha + \sqrt{-3\,\alpha^2 + 4\,\alpha}}{\alpha}$$

```
> p1<1-p2;
```

$$0 < \frac{\alpha - 1}{\alpha}$$

```
> 1-p2<2;
```

$$\frac{\alpha - 1}{\alpha} < 1$$

Numerical simulations

```
> alpha:=0.5;beta:=0.2;N:=30;
```

$$\alpha := 0.5$$

$$\beta := 0.2$$

$$N := 30$$

```
> f(u,v);
```

$$\frac{0.5\, u}{1 + 0.2\, v}$$

```
> x[0]:=20;x[1]:=40;
```

$$x_0 := 20$$

$$x_1 := 40$$

```
> for i from 0 to N-2 do
    x[i+2]:=f(x[i+1],x[i])
```

```
end do:
> plot([[n,x[n]]$n=0..N],style=point,symbol=circle);
```




```
> alpha:=1.5;beta:=0.2;N:=30;
```

$$\alpha := 1.5$$

$$\beta := 0.2$$

$$N := 30$$

```
> f(u,v);
```

$$\frac{1.5\,u}{1 + 0.2\,v}$$

```
> x1:=(alpha-1)/beta;
```

$$x1 := 2.500000000$$

```
> x[0]:=4;x[1]:=1;
```

$$x_0 := 4$$

$$x_1 := 1$$

```
> for i from 0 to N-2 do
    x[i+2]:=f(x[i+1],x[i])
  end do:
> plot([[[n,x[n]]$n=0..N],[[n,x1]$n=0..N]],style=[point,point],
  symbol=[circle,cross],color=[red,green]);
```



4
3
2
1
0
10
20
30