Modelling with first order differential equations

```
> restart:with(DEtools):
```

Radioactive decay

Ratherford Law: The rate of decay for radioactive material is proportional to the number of atoms present

$x(t)$ - the amount of radioactive material at the time t
x_0 - the amount of radioactive material at the initial time t_0 ($t_0 = 0$)

$\frac{dx}{dt}$ = x'(t) represent the rate of decay

The model:
x'(t) = $-k\,x(t)$
$x(0) = x_0$

k-decay constant

```
> RD_eq:=diff(x(t),t)=-k*x(t);
```

$$RD\_eq := \frac{\mathrm{d}}{\mathrm{d}t}\,x(t) = -k\,x(t)$$

```
> sol:=dsolve({RD_eq,x(0)=x0},x(t));
```

$$sol := x(t) = x0\,\mathrm{e}^{-k\,t}$$

```
> x_sol:=unapply(rhs(sol),t,x0,k);
```

$$x\_sol := (t, x0, k) \rightarrow x0\,\mathrm{e}^{-k\,t}$$

```
> plot([x_sol(t,2,1/8000),x_sol(t,4,1/8000),x_sol(t,2,1/10000)
  ],t=0..100000);
```



The ***half-life*** time of a radioactive substance is the lenght of time it takes the material to decay to half of its original amount.

the half-life time can be calculated from the ecuation $x\left(T_{\frac{1}{2}}\right) = \frac{x_0}{2}$

```
> eq:=x_sol(T12,x0,k)=x0/2;
```

$$eq := x0\, e^{-k\, T12} = \frac{1}{2}\, x0$$

```
> T12=solve(eq,T12);
```

$$T12 = \frac{\ln(2)}{k}$$

Usualy the rate of decay for radioactive material is given in the terms of the half-life time, so, if we want to apply this model first we have to find the decay constant k.

For example, in the case radioactive carbon isotope C^{14}, the half-life time is 5730 years, then the corresponding decay constant is

```
> T12_C14:=5730;
```

$$T12\_C14 := 5730$$

```
> k_C14:=ln(2)/T12_C14;
```

$$k\_C14 := \frac{1}{5730}\, \ln(2)$$

```
> evalf(k_C14);
```

$$0.0001209680943$$

```
> plot(x_sol(t,200,k_C14),t=0..100000);
```



C^{14} Radiocarbon Dating method

An important tool in archeological research is radiocarbon dating. This is a means of determining the age of certain wood and plant remains, hence of animal or human bones or artifacts found buried at the same levels. The procedure was developed by the American chemist Willard Libby (1908–1980) in the early 1950s and resulted in his winning the Nobel prize for chemistry in 1960.

Radiocarbon dating is based on the fact that somewood or plant remains contain residual amounts of carbon-14, a radioactive isotope of carbon. This isotope is accumulated during the lifetime of the plant and begins to decay at its death. Since the half-life of carbon-14 is long (approximately 5730 years), measurable amounts of carbon-14 remain after many thousands of years. Libby showed that if even a tiny fraction of the original amount of carbon-14 is still present, then by appropriate laboratory measurements the proportion of the original amount of carbon-14 that remains can be accurately determined. In other words, if x(t) is the amount of carbon-14 at time t and x0 is the original amount, then the ratio x(t)/x0 can be determined, at least if this quantity is not too small. Present measurement techniques permit the use of this method for time periods up to about 50,000 years, after which the amount of carbon-14 remaining is only about 0.00236 of the original amount.

```
> RD_eq;
```

$$\frac{d}{dt}x(t) = -k\,x(t)$$

```
> x_sol(t,x0,k);
```

$$x0\,e^{-kt}$$

Suppose that at the discovering time T the amount of rezidual has the value x_1 then the value of T can be found it from the equation
$x(T) = x_1$

```
> eq:=x_sol(T,x0,k_C14)=x1;
```

$$eq := x0\,e^{-\frac{1}{5730}\ln(2)\,T} = x1$$

```
> T=solve(eq,T);
```

$$T = -\frac{5730\ln\left(\frac{x1}{x0}\right)}{\ln(2)}$$

Usualy, the value of x_1 is given in procent from original amount x_0. For example, Suppose that certain remains are discovered in which the current residual amount of carbon-14 is 20% of the original amount. Determine the age of these remains.

```
> p:=20/100;
```

$$p := \frac{1}{5}$$

```
> x1:=p*x0;
```

$$x1 := \frac{1}{5}\,x0$$

```
> eq:=x_sol(T,x0,k_C14)=x1;
```

$$eq := x0\,e^{-\frac{1}{5730}\ln(2)\,T} = \frac{1}{5}\,x0$$

```
> Tf:=solve(eq,T);
```

$$Tf := \frac{5730\ln(5)}{\ln(2)}$$

```
> evalf(Tf);
```

$$13304.64798$$

Thermal Cooling

Newton's law of cooling. If *T(t)* is the temperature of an object at time t and T_{out} is the temperature of its surroundings, then the change rate of the surface temperature of an object is proportion with the difference between the object temperature *T(t)* and the surrounding temperature T_{out}

$$\frac{dT}{dt} = -k\,(T - T_{out})$$
$$T(0) = T_0$$
where k is a positive constant called the cooling coefficient.

This becomes a "law of warming" if the surroundings are hotter than the object.

```
> N_eq:=diff(T(t),t)=-k*(T(t)-T_out);
```

$$N\_eq := \frac{d}{dt}\,T(t) = -k\,(T(t) - T\_out)$$

```
> sol:=dsolve({N_eq,T(0)=T0},T(t));
```

$$sol := T(t) = T\_out + e^{-k t}\,(T0 - T\_out)$$

```
> T_sol:=unapply(rhs(sol),t,k,T0,T_out);
```

$$T\_sol := (t, k, T0, T\_out) \rightarrow T\_out + e^{-k t}\,(T0 - T\_out)$$

```
> plot([T_sol(t,2,30,15),T_sol(t,2,10,15),T_sol(t,2,15,15)],t=
  0..5);
```



Exercise: How long will it take for a 100^{o} C egg to cool to 60^{o} C in a 21^{o} C room if k = 0.03419 $\min^{-1}$?

we have the solution of Newton cooling model

```
> T_sol(t,k,T0,T_out);
```

$$T\_out + e^{-k t} (T0 - T\_out)$$

```
> T0:=100;T_out:=21;k:=0.03419;Tf:=60;
```

$$T0 := 100$$
$$T\_out := 21$$
$$k := 0.03419$$
$$Tf := 60$$

The answer can be obtained from the equation
$T(time) = T_{final}$

```
> eq:=T_sol(t1,k,T0,T_out)=Tf;
```

$$eq := 21 + 79\,\mathrm{e}^{-0.03419\,t1} = 60$$

```
> t1=solve(eq,t1);
```

$$t1 = 20.64598439$$

Time-Dependent Outside Temperature

When considering the cooling model is separable because T_{out} is constant in this instance. Let's consider what happens when the outside temperature changes with time.
We can still use Newton's law of cooling, so that if *T(t)* is the object temperature and $T_{out}(t)$ is the room's temperature, then

$$\frac{dT}{dt} = -k\,\left(T(t) - T_{out}(t)\right)$$
$$T(0) = T_0$$

Note that this equation is not separable (because T_{out} varies with time) but it is linear, so we can find its general solution as follows. Rearrange the terms to give the linear ODE in standard form:

$$\frac{dT}{dt} + k\,T(t) = k\,T_{out}(t)$$
$$T(0) = T_0$$

```
> restart;with(DEtools):
> N_eq_var:=diff(T(t),t)=-k*(T(t)-T_out(t));
```

$$N\_eq\_var := \frac{\mathrm{d}}{\mathrm{d}t}\,T(t) = -k\,\left(T(t) - T\_out(t)\right)$$

```
> sol_var:=dsolve({N_eq_var,T(0)=T0},T(t));
```

$$sol\_var := T(t) = \left(\int_0^t k\,T\_out(\_z1)\,\mathrm{e}^{k\,\_z1}\,\mathrm{d}\_z1 + T0\right)\mathrm{e}^{-k\,t}$$

Exercise: Find the solution in the case of $T_{out}(t) = 21 - 5\,\sin\left(\frac{2\,\pi\,(t+3)}{24}\right)$ in the case of the egg problem

```
> T_out:=t->21-5*sin(2*Pi*(t+3)/24);
```

$$T\_out := t \rightarrow 21 - 5\,\sin\left(\frac{1}{12}\,\pi\,(t+3)\right)$$

```
> plot(T_out(t),t=0..24);
```




```
> N_eq_var;
```

$$\frac{d}{dt} T(t) = -k \left(T(t) - 21 + 5 \sin\left(\frac{1}{12} \pi (t+3) \right) \right)$$

```
> T0:=100;k:=0.03419;
```

$$T0 := 100$$

$$k := 0.03419$$

```
> sol_var:=dsolve({N_eq_var,T(0)=T0},T(t));
```

$$sol\_var := T(t) = e^{-\frac{3419}{100000}t}\left(100 - \frac{3\left(213687500\,\pi\sqrt{2} + 4375000000\,\pi^2 - \frac{175343415}{2}\sqrt{2} + 736442343\right)}{625000000\,\pi^2 + 105206049}\right) + \frac{1}{625000000\,\pi^2 + 105206049}\left(3\left(427375000\,\pi\cos\left(\frac{1}{12}\pi t + \frac{1}{4}\pi\right) + 4375000000\,\pi^2 - 175343415\sin\left(\frac{1}{12}\pi t + \frac{1}{4}\pi\right) + 736442343\right)\right)$$

```
> T_sol1:=unapply(rhs(sol_var),t);
```

$$T\_sol1 := t \rightarrow e^{-\frac{3419}{100000}t}\left(100 - \frac{3\left(213687500\,\pi\sqrt{2} + 4375000000\,\pi^2 - \frac{175343415}{2}\sqrt{2} + 736442343\right)}{625000000\,\pi^2 + 105206049}\right) + \frac{1}{625000000\,\pi^2 + 105206049}\left(3\left(427375000\,\pi\cos\left(\frac{1}{12}\pi t + \frac{1}{4}\pi\right) + 4375000000\,\pi^2 - 175343415\sin\left(\frac{1}{12}\pi t + \frac{1}{4}\pi\right) + 736442343\right)\right)$$

```
> plot([T_sol1(t),T_out(t)],t=0..4*60,color=[red,blue]);
```



Air Conditioning a Room

Now let’s build a model that describes a room cooled by an air conditioner. Without air conditioning, we can model the change in temperature using the Newton model. When the air conditioner is running, its coils remove heat energy at a rate proportional to the difference between $T_r(t)$, the room temperature, and the temperature T_{ac} of the coils. So, using Newton’s law of cooling for the temperature change due to both the air outside the room and the air conditioner coils, our model ODE is

$$\frac{dT_r}{dt} = -k\left(T_r(t) - T_{out}\right) - k_{ac}\left(T_r(t) - T_{ac}\right)$$
$$T_r(0) = T_0$$

where T_{out} is the temperature of the outside air and k and k_{ac} are the appropriate cooling coefficients. If the unit is turned off, then $k_{ac} = 0$ and this equation reduces to Newton equation.

```
> restart:
> deq:=diff(T_r(t),t)=-k*(T_r(t)-T_out)-k2*(T_r(t)-T_ac);
```

$$deq := \frac{d}{dt}\, T\_r(t) = -k\,(T\_r(t) - T\_out) - k2\,(T\_r(t) - T\_ac)$$

```
> dsolve({deq,T_r(0)=T0},T_r(t));
```

$$T\_r(t) = e^{-(k + k2)\, t}\left(T0 - \frac{T\_ac\; k2 + T\_out\; k}{k + k2}\right) + \frac{T\_ac\; k2 + T\_out\; k}{k + k2}$$

Let's assume that the initial temperature of the room is 15^o C and the outside temperature is a constant 35^o C. The air conditioner operates with a coil temperature of 5^o C, k = 0.03 $\min^{-1}$ and $k_{ac} = 0.1 \min^{-1}$, then the solution is

```
> T0:=15;T_out:=35;T_ac:=5;k:=0.03;k2:=0.1;
```

$$T0 := 15$$
$$T\_out := 35$$
$$T\_ac := 5$$
$$k := 0.03$$
$$k2 := 0.1$$

```
> sol:=dsolve({deq,T_r(0)=T0},T_r(t));
```

$$sol := T\_r(t) = \frac{155}{13} + \frac{40}{13}\, e^{-\frac{13}{100}\, t}$$

```
> T_r_sol:=unapply(rhs(sol),t);
```

$$T\_r\_sol := t \rightarrow \frac{155}{13} + \frac{40}{13}\, e^{-\frac{13}{100}\, t}$$

```
> evalf(155/13);
```

$$11.92307692$$

```
> plot(T_r_sol(t),t=0..60);
```



Let's complicate the problem. The air conditioner switches on when the room reaches 25^o C, and switches off at 20^o C.

At some time t[on] the room's temperature will reach 25^o C and the air conditioner will switch on. For t > t_{on} the temperature is modeled by the IVP

$$\frac{dT_r}{dt} = -k\left(T_r(t) - T_{out}\right) - k_{ac}\left(T_r(t) - T_{ac}\right)$$

$$T_r(0) = T_0$$

with $T_0 = 25$, $T_{out} = 35$, $k = 0.03\ \mathrm{min}^{-1}$ and $k_{ac} = 0.1\ \mathrm{min}^{-1}$, which is valid until the room cools to 20^o C at some time t_{off}.

```
> T0:=25;
```

$$T0 := 25$$

```
> deq;
```

$$\frac{d}{dt} T\_r(t) = -0.13\, T\_r(t) + 1.55$$

```
> sol1:=dsolve({deq,T_r(0)=T0},T_r(t));
```

$$sol1 := T\_r(t) = \frac{155}{13} + \frac{170}{13} e^{-\frac{13}{100} t}$$

```
> T_on:=unapply(rhs(sol1),t);
```

$$T\_on := t \to \frac{155}{13} + \frac{170}{13} e^{-\frac{13}{100} t}$$

t_{off} is found from the equation $T\_on(t_{off}) = 20$

```
> eq:=T_on(t_off)=20;
```

$$eq := \frac{155}{13} + \frac{170}{13} e^{-\frac{13}{100} t\_off} = 20$$

```
> t_off:=solve(eq,t_off);
```

$$t\_off := -\frac{100}{13} \ln\left(\frac{21}{34}\right)$$

```
> evalf(t_off);
```

$$3.706446822$$

For $t > t_{off}$ the air conditioner is turned off which means $k_{ac} = 0$ and we have the solution

```
> deq1:=diff(T_r(t),t)=-k*(T_r(t)-T_out);
```

$$deq1 := \frac{d}{dt} T\_r(t) = -0.03\, T\_r(t) + 1.05$$

```
> sol2:=dsolve({deq1,T_r(t_off)=20},T_r(t));
```

$$sol2 := T\_r(t) = 35 - \frac{5}{7} e^{-\frac{3}{100} t}\, 21^{10/13}\, 34^{3/13}$$

```
> T_off:=unapply(rhs(sol2),t);
```

$$T\_off := t \to 35 - \frac{5}{7} e^{-\frac{3}{100} t}\, 21^{10/13}\, 34^{3/13}$$

```
> eq1:=T_off(t_on)=25;
```

$$eq1 := 35 - \frac{5}{7} e^{-\frac{3}{100} t\_on}\, 21^{10/13}\, 34^{3/13} = 25$$

```
> t_on:=solve(eq1,t_on);
```

$$t\_on := -\frac{100}{39} \ln\left(\frac{351232}{290107737}\right)$$

```
> evalf(%);
```

$$17.22195042$$

```
> T_sol:=t->piecewise(0<=t and t<t_off,T_on(t),t_off<=t and t<=
  t_on,T_off(t));
```

$$T\_sol := t \rightarrow piecewise(0 \le t \textbf{ and } t < t\_off, T\_on(t), t\_off \le t \textbf{ and } t \le t\_on, T\_off(t))$$

```
> evalf(T_sol(t_on));
```

$$25.00000000$$

```
> T_off(t_off);
```

$$20$$

```
> plot(T_sol(t),t=0..t_on);
```


25
24
23
22
21
0
2
4
6
8
10
12
14
16
t